Delay Plot GNUPlot Output and Code

Alex Striff

August 26, 2018



Figure 1: Paper, color version.



Figure 2: Paper, grayscale version.



Figure 3: Dark Owl version (beamer slides).

```
set terminal pngcairo transparent enhanced font "Droid Sans,72" \
      fontscale 1.0 size 3200, 2400

# Dark Owl
text = '#ffffff'
shade = '#5d5a96' # (rgb(OwlRed) .+ rgb(OwlBlue)) .* 3/8
mindelay = '#f29318' # OwlYellow
maxdelay = '#ff5ca8' # OwlRed
set output 'delayplot-do.png'

# Paper Color
# text = '#000000'
# shade = '#bfbce4' # 0.25*OwlRed + 0.25*OwlBlue + 0.5*white
# mindelay = '#f29318' # OwlYellow
# maxdelay = '#ff5ca8' # OwlRed
# set output 'delayplot-pc.png'

# Paper Grayscale
# text = '#000000'
```

```
# shade = '#c0c0c0'
# mindelay = '#000000'
# maxdelay = '#000000'
# set output 'delayplot-pg.png'

set multiplot
set style increment default

set border lw 3 lc rgb text
set key tc rgb text
set xlabel tc rgb text
set ylabel tc rgb text

set datafile separator ','
dlin = 'delayplot.csv'

set origin 0,0
set size  1,1
set xtics auto
set key left top autotitle columnhead
set xlabel 'Words'
set ylabel ' (ms)' offset 1.5,0
set logscale xy

plot [10:3e5][] \
       dlin u 3:2:1 w filledcurves lt rgb shade t '', \
       dlin u 3:1 w linespoints pt 9 ps 3 lw 4 lc rgb mindelay t '', \
       dlin u 3:2 w linespoints pt 7 ps 3 lw 4 lc rgb maxdelay t '', \
       NaN w linespoints pt 9 ps 3 lw 4 lc rgb mindelay t 'Minimum delay', \
       NaN w linespoints pt 7 ps 3 lw 4 lc rgb maxdelay t 'Maximum delay'

set origin 0.5,0.15
set size  0.45,0.4
set xtics 11, 5
set ytics 0.1
unset key
set xlabel 'Words' offset 0,0.325
set ylabel ' (ms)' offset 1.5,0
unset logscale

plot [10:21][0.15:0.5] \
       dlin u 3:2:1 w filledcurves lc rgb shade, \
       dlin u 3:1 w linespoints pt 9 ps 3 lw 4 lc rgb mindelay, \
       dlin u 3:2 w linespoints pt 7 ps 3 lw 4 lc rgb maxdelay

unset multiplot
```

Difference Plot GNUPlot Output and Code

Alex Striff

August 26, 2018



Figure 1: Paper, color version.



Figure 2: Paper, grayscale version.



Figure 3: Dark Owl version (beamer slides).

```
set terminal pngcairo transparent enhanced font "Droid Sans,72" \
      fontscale 1.0 size 3600, 2400

# Dark Owl
# text     = '#ffffff'
# aout     = '#f29318' # OwlYellow
# abessel4 = '#ff5ca8' # OwlRed
# adiff    = '#0098e9' # OwlBlue
# horiz    = '#a29bff' # (rgb(OwlRed) .+ rgb(OwlBlue)) norm. to max. blue
# set output 'diffplot-do.png'

# Paper Color
text     = '#000000'
aout     = '#f29318' # OwlYellow
abessel4 = '#ff5ca8' # OwlRed
adiff    = '#0098e9' # OwlBlue
horiz    = '#a29bff' # (rgb(OwlRed) .+ rgb(OwlBlue)) norm. to max. blue
set output 'diffplot-pc.png'

# Paper Grayscale
# text     = '#000000'
```

```
# aout     = '#000000'
# abessel4 = '#888888'
# adiff    = '#000000'
# horiz    = '#ffffff'
# set output 'diffplot-pg.png'

set border lw 3 lc rgb text
set key tc rgb text
set xlabel tc rgb text
set ylabel tc rgb text

set multiplot
set style increment default

set datafile separator ','
paout  = 'paout.csv'
padiff = 'padiff.csv'

# Difference
set size   1, 0.5
set origin 0, 0
set lmargin 8
set xtics auto
unset key
set xlabel 'Time (s)'
set ylabel 'Voltage (mV)'
set xrange [105:295] noreverse writeback
# set xrange [199:201] noreverse writeback # Faster to plot for debugging

plot \
       padiff u 1:2 w lines lw 2 lt rgb adiff, \
       padiff u 1:3 w lines lw 5 lt rgb horiz, \
       padiff u 1:($3 + $4) w lines lw 5 lt rgb horiz, \
       padiff u 1:($3 - $4) w lines lw 5 lt rgb horiz

# Waveforms zoom-in
set size   0.25, 0.25
set origin 0.695, 0.56
set lmargin 0
set xtics auto 0.25
set ytics auto 0.1
unset key
set xlabel 't (s)' offset 0,0.5
set ylabel 'V (V)' offset 1.25,0
set xrange [199.75:200.25] noreverse writeback

plot \
       paout u 3:($4 * 1e-3) w lines lw 4 lt rgb aout t '', \
```

```
       paout u 5:($6 * 1e-3) w lines lw 4 lt rgb abessel4 t ''

# Waveforms
set size   1, 0.5
set origin 0, 0.5
set lmargin 8
set xtics auto format ''
set ytics auto
unset xlabel
set key left top autotitle columnhead
set ylabel 'Voltage (V)'
set xrange [105:295] noreverse writeback
# set xrange [199:201] noreverse writeback # Faster to plot for debugging

plot \
       paout u 3:($4 * 1e-3) w lines lw 4 lt rgb aout t 'Output', \
       paout u 5:($6 * 1e-3) w lines lw 4 lt rgb abessel4 t 'Filtered Output'

unset multiplot
```

Triangle/Spumpus Plot GNUPlot Output and Code

Alex Striff

August 26, 2018



Figure 1: Paper, color version.



Figure 2: Paper, grayscale version.



Figure 3: Dark Owl version (beamer slides).

```
set terminal pngcairo transparent enhanced font "Droid Sans,72" \
      fontscale 1.0 size 3600, 2400

# Dark Owl
text = '#ffffff'
spcolor = '#0098e9' # OwlBlue
set output 'spumpus-do.png'

# Paper Color
# text = '#000000'
# spcolor = '#0098e9' # OwlBlue
# set output 'spumpus-pc.png'

# Paper Grayscale
# text = '#000000'
# spcolor = '#000000'
# set output 'spumpus-pg.png'

set style increment default

set border lw 3 lc rgb text
```

```
set xlabel tc rgb text
set ylabel tc rgb text

set datafile separator ','
sphist = 'spumpus-hist.csv'
spout  = 'spumpus-out.csv'

unset key
set xlabel 'Time (ms)'
set ylabel 'Voltage (V)'
set xrange [0:3] noreverse writeback

plot \
       sphist u ($1 * 1e3):($2) w lines lw 4 lt rgb spcolor
```